
8. Рішення диференційних рівнянь
8.1.Головні засоби вирішення диференційних рівнянь. Пакет рішення диференційних рівнянь DEtools. 

8.2. Графічне представлення рішень. 

8.3. Поглиблений аналіз диференційних рівнянь.

8.1. Головні засоби вирішення диференційних рівнянь. Пакет рішення диференційних рівнянь DEtools

> restart;
Головні засоби вирішення диференційних рівнянь.

Важливе місце в математичних розрахунках займає рішення диференційних рівнянь. До нього, зокрема, звичайно відноситься аналіз поведінки різних систем в часі (аналіз динаміки), а також обчислення різних полів (поля тяжіння, полів електричних зарядів тощо). Важко переоцінити роль диференційних рівнянь в моделюванні фізичних і технічних об'єктів і систем, Maple  дозволяє вирішувати одиночні диференційні рівняння і системи диференційних рівнянь як аналітично, так і у числовому вигляді. Розробниками системи оголошено про істотне розширення засобів рішення диференційних рівнянь і про підвищення їх надійності в значенні знаходження рішень для більшості класів диференційних рівнянь. Для вирішення одиночних простих диференційних рівнянь (задача Коші) використовується функція dsolve в різних формах запису: 

dsolve(ODE)  
dsolve(ODE, y(x), extra_args)  
dsolve({ODE, ICs}, y(x), extra_args)
Тут ODE - звичайне диференційне рівняння, y(x) - функція однієї змінної, ICs- початкові умови, extra_args - необов`язкова опція, яка задає тип рішення диференційного рівняння. Відзначимо основні значення цього параметра:  

exact — аналітичне рішення (прийнято за умовчанням);  

explicit — рішення в явному вигляді;  

system — рішення системи диференційних рівнянь;  

ICs — рішення системи диференційних рівнянь із заданими початковими умовами;  

formal series — рішення у формі поліному;  

integral transform — рішення на основі інтегральних перетворень Лапласа, Фурье тощо.;  

series — рішення у вигляді ряду з порядком, який вказується значенням змінної Order; 

numeric — рішення в чисельному вигляді. 

 Давайте почнемо з того, щоб отримати  рішення  протсого диференційного рівняння, яке Ви ймовірно бачили в курсі загальної фізики, а саме рівняння для радіоактивного розпаду ядер.  Якщо стала розпаду атомів в зразку є [image: image1.wmf]l

 ([image: image2.wmf]l

 має розмірність 1/с), то рівняння, що визначає швидкість розпаду ядер, є наступним:

> ODE1:=diff(N(t),t)=-lambda*N(t);

З початковою умовою, що у момент часу [image: image4.wmf] = 

 кількість ядер складає [image: image5.wmf]N

0

. Якщо потрібно знайти саме залежність кількості ядер, які ще не розпалися від часу, то використовуємо команду:

> dsolve(ODE1);
[image: image6.wmf] = 

Як слід було очікувати, рішення диференційного рівняння містить невизначену константу інтегрування [image: image7.wmf]_C1

 через те, що ми ніяк не визначили початкових умов. Якщо ми хотіли би задати їх, то слід було виористати дещо інший синтаксис команди dsolve:

> dsolve({ODE1,N(0)=N[0]},N(t));

Або, наприклад, у такій формі:

> Pochatkova_Umova:=N(0)=N[0]:

dsolve({ODE1,Pochatkova_Umova},N(t)); 

Диференційні рівняння другого порядку надзвичайно поширені у фізиці, хоча б тому, що другий закон Ньютона є диференційним рівнянням другого порядку, так само як і рівняння дифузії, хвильове рівняння, рівняння Шредінгера в квантовій механіці, тощо. 

Запишемо таке диференційне рівняння для випадку вертикального руху матеріальної точки у полі сил тяжіння Землі (тоді [image: image10.wmf]g

 - прискорення ввільного падіння).  З другого закону Ньютона отримаємо, що прискорення [image: image11.wmf]d

  дорівнює [image: image12.wmf]-

, якщо вісь [image: image13.wmf]Oy

  спрямована вгору:

> ode2:=diff(y(t),t$2)=-g;
[image: image14.wmf] := 

Випишемо початкові умови:

> PU:=y(0)=y[0],D(y)(0)=v[0];
[image: image15.wmf] := 

> dsolve({ode2,PU},y(t));
[image: image16.wmf] = 

Зверніть увагу, по-переше ми використали оператор похідної [image: image17.wmf](

, після чого підставили у цей оператор ([image: image19.wmf]0

), тобто створили форму [image: image20.wmf](

, яка означає, що похідна береться в точці [image: image21.wmf] = 

Ось ще одне выдоме рівняння другого порядку, рівняння гармонічного осцилятора:

> ode3:=diff(x(t),t$2)+omega^2*x(t)=0;
[image: image22.wmf] := 

Maple вирішує це рівняння в його загальному вигляді лінійної комбінації двох гаромнічних функцій  із двома невизначеними константами інтегрування:

> dsolve(ode3,x(t));
[image: image23.wmf] = 

Часто більш зручно мати справу з системами диференційних рівнянь, аніж з одиночними диференційними рівняннями другого, третього, або ще вищого порядку. Можна записати наше диференційне рівняння високого порядку як систему диференційних рівнянь першого порядку.

Для іллюстрації є простий приклад.  Якщо ми позначимо dv/dt як прискорення, а dx/dt як швидкість, то рівняння другого порядку, яке ми щойно вирішили для лінійного гармонічного осцилятора може бути записано, як наступна система рівнянь першого порядку:

> Eq1:=diff(x(t),t)=v(t);

Eq2:=diff(v(t),t)=-omega^2*x(t);
[image: image24.wmf] := 

[image: image25.wmf] := 

І далі вирішимо цю систему з двох диференційних рівнянь першого порядку:

> dsolve({Eq1,Eq2},{v(t),x(t)});
[image: image26.wmf]{

 Пакет рішення диференційних рівнянь DEtools. 
Рішення диференційних рівнянь найрізноманітніших типів — одна з переваг системи Maple . Пакет DEtools надає великий набір корисних функцій для вирішення диференційних рівнянь і систем з такими рівняннями. Пакет підключається таким способом:

> with (DEtools);
Загальна кількість функцій та команд у цьому пакеті перевищує можливості будь-якої лекції. Зупинимося побіжно лише на описові деяких з них:

DEnormal — повертає нормалізовану форму диференційних рівнянь;  

DEplot — будує графіки рішення диференційних рівнянь;   

DEplot3d — будує тривимірні графіки для рішеннь систем диференційних рівнянь;   

Dchangevar — зміна перемінних у диференційних рівняннях;  

PDEchangecoords — зміна координатних систем для диференційних рівнянь у часткових похідних;   

PDEplot — побудова графіків рішення диференаціальних рівнянь у часткових похідних;  

autonomous — тестує диференційні рівняння на автономність;  

convertAlg — повертає список коефіцієнтів для диференційних рівнянь;  

convertsys — перетворить систему диференційних рівнянь в систему одиночних рівнянь;  

dfieldplot — будує графік рішень диференційних рівнянь у вигляді векторного поля;  

indicialeq — перетворить диференційні рівняння у поліноміальні;  

phaseportrait — будує графік рішення диференційних рівнянь у формі фазового портрету;  

reduceOrder — знижує порядок диференційних рівнянь;  

regularsp — обчислює регулярні особливі точки для диференційних рівнянь другого порядку; 

translate — перетворить диференційні рівняння в список операторів;  

untranslate — перетворить список операторів в диференційні рівняння;  

varparam — знаходить загальне рішення диференційних рівнянь методом варіації параметрів. 

Решта функцій пакету DEtools детальніше описується у довідковій системі Maple.
8.2.  Графічне представлення рішень
Maple  в змозі вирішувати і складніші рівняння, ніж ті простенькі фізичні задачі, які ми розглянули вище. Ускладнимо задачу про гармонічний осцилятор і врахуємо опір в системі шляхом запровадження сили опору: [image: image27.wmf] = 

 - час релаксації осцилятора, а [image: image29.wmf]mv

 - його імпульс. Диференційне рівняння згасаючих коливань, як відомо, може бути записане у формі:

> restart:Eqn:=diff(x(t),t$2) + diff(x(t),t)/tau + omega^2*x(t)=0;
[image: image30.wmf] := 

> dsolve(Eqn,x(t));

Рішення виглядає доволі солідно. Втім ми нібито вирішували задачу про осцилятор з демпфером (опором), то де ж поділися наші періодичні функції - сінус і косінус?  Проблема в тому, що Maple не знає (так само як і Ви) як співвідносяться поміж собою і якими є параметри рівняння: [image: image32.wmf],

w

t

Давайте думати як фізики: припустимо, що циклічна частота приблизно дорівнює [image: image33.wmf]2

, отже період вільних коливань [image: image34.wmf] = 

 складає 1 секунду. Далі припустимо, що осцилятор (нехай це звичайний  маятник) знаходиться у моторному маслі при  температурі у 10 градусах нижче нуля. Таким чином, що характерний час релаксації складає всього 0.05 секунди (в'язке тертя настільки сильне, що зупинка руху відбувається за 0.05 секунди). 

Ваша інтуїція повинна повідомити Вам, що якщо ви відхиляєте такий маятник і відпускаєте його, ви не сподіваєтесь побачити декілька його повних качань: все що може трапитися - це те, що маятник поволі повертатиметьсяся у вертикальне положення і залишаться далі в ньому. Давайте подивимось, що з цього приводу каже Maple:

> omega:=2*Pi; tau:=0.05;
[image: image35.wmf] := 

> X:=dsolve({Eqn,x(0)=1,D(x)(0)=0},x(t));
[image: image37.wmf] := 

· plot(rhs(X),t=0..8, style=point, color=navy, numpoints=40);

[image: image38.png]



Maple показує, що він згідний з Вашою інтуїцією: рух маятника в цьому випадку є аперіодичним, експоненціально згасаючим рухом до положення рівноваги. Зверніть увагу на команду [image: image39.wmf](

 у тілі команди plot: ця команда пропонує створити графік правої частини виразу X, а в правій частині цього виразу саме й стоїть рішення нашого диференційного рівняння. Чи те саме вийде, якщо ми схочемо побудувати графік лівої частини ([image: image40.wmf](

Тепер спробуємо змінити опір руху маятника і подивитися, що з того вийде. Ваша інтуїція знов підкаже, що якщо масло підігріти, або взагалі винести маятник у повітря, тобто збільшити час релаксації так, що [image: image41.wmf] < 

, то маятник буде качатися, хіба що повільно зменшуючи амплітуду. Для прикладу, припустимо, що [image: image42.wmf] = 

> omega:=2*Pi; tau:=5;
[image: image43.wmf] := 

> X:=dsolve({Eqn,x(0)=1,D(x)(0)=0},x(t));
[image: image45.wmf] := 

· plot(rhs(X),t=0..8,  color=navy, numpoints=500);

[image: image46.png]05

Як бачимо, тепер маятник коливається, хоча амплітуда при цьому повільно зменшується. В рішенні ж з'явилися періодичні функції сінус і косінус. Секрет у тому, що існує формула Ейлера для експоненти:

щойно показник експоненти у рішенні диференційного рівняння стає уявним, так саме рішення становиться осцилюючим.

У прикладі вивчення згасаючих коливань ми користувалися для графічного зображення рішень звичайною функцією візуалізації plot - це найбільш простий і природній засіб  графічного представлення рішень.

Втім у Maple є також спеціальні інструменти для графічного відтворення пішень диференційних рівнянь. Найбільш поширеною є команда odeplot з пакету plots. Підключимо цей пакет так:

> with(plots);
Warning, the name changecoords has been redefined

Серед переліку команд цього пакету команд є і команда odeplot  :
odeplot(dsn, vars, range, options)
де dsn - вихідна форма числового рішення диференційного рівняння, vars-лист (перелік) змінних та функцій у рішенні, range- визначення інтервалу, на якому будується графік, options- звичайні опції 2D-графіки.

Застосуємо цю функцію до вирішеної вже задачі. Перш за все отримаємо рішення, яке раніше отримували в аналітичному вигляді, у числовому форматі:

> X:=dsolve({Eqn,x(0)=1,D(x)(0)=0},x(t),numeric);
[image: image48.wmf] := 

На перший погляд здається, що ми просто переписали вже одного разу отримане  рішення. Проте, це не так, адже ми  додали важливу опцію  numeric, отже задали вихід у числовій , а не в аналітичній формі. Вона виглядає не дуже прозоро, аналітичне рішення було значно зрозумілішим, а вітм  для графічного виводу це не перешкода.

До речі дуже важливо те, що і в тих випадках, коли  рішення не може бути отримане в аналітичному вигляді, воно завжди можливе у числовому форматі. Саме такий формат і використовує функція візуалізації рішення odeplot: 

· odeplot(X,[t,x(t)],0..8,color=navy,numpoints=400);

[image: image49.png]



Отриманий таким шляхом графік нічим не гірший  графіку аналітичного рішення, наведеного вище. За виключенням того, що аналітичне рішення існує не завжди, тоді як числове рішення можливе в кожному випадку.

Використання комп'ютерів в останні 30-40 років сприяло повному перегляду старих надважких задач класичної механіки. Система трьох диференційних рівнянь, яку ми зараз запишемо, винайдена Едвардом Лоренцем, який намагався моделювати... погоду:

> restart;
> Eq1:=diff(x(t),t) = sigma*(y(t)-x(t));
> Eq2:=diff(y(t),t) = r*x(t)-y(t)-x(t)*z(t);
> Eq3:=diff(z(t),t) = x(t)*y(t)-b*z(t);

[image: image52.wmf] := 

де константи [image: image53.wmf]s

, [image: image54.wmf]r

  та  [image: image55.wmf]b

 можна обирати, як Вам заманеться. Для більшості наборів цих параметрів рішення наведної системи є не дуже цікавими, але Лоренц відкрив, що є ряд таких наборів цих трьох параметрів, для яких рішення поводяться надзвичайно і незвично.  Якщо ми вибираємо

> sigma:=10;b:=8/3;
[image: image56.wmf] := 

і далі змінюємо параметр [image: image58.wmf]r

  від 0 до приблизно 30 відбувається стрімкий перехід  від звичайної і нудної поведінки рішень наведеної системи рівнянь до такої незвичної і дивної, що цей факт революціонізував багато областей науки, включно з фізикою.  Щоб на конкретному прикладі прослідкувати цей перехід візуально, для конкретного вибору r, який ми використовуємо, знайдемо числове рішення системи:

> r:=10;s:=dsolve({Eq1,Eq2,Eq3,x(0)=1,y(0)=1,z(0)=1},{x(t),y(t),z(t)},type=numeric);
[image: image59.wmf] := 

і щоби подивитися на що схоже це рішення застосуємо plots[odeplot]:
> plots[odeplot](s,[[t,x(t)],[t,y(t)],[t,z(t)]],0..20,numpoints=200, thickness=3, tickmarks=[3,4]);

[image: image61.png]20




Не пошкодуйте пари хвилин і "проклацайте" наведені командні рядки з наступними значеннями параметру [image: image62.wmf]r

 : 10,  20,  28,  35, 45. Для r трохи більше 25  більш менш регулярні коливання формують неповторні ексцеси, такі собі "вибрики".  І що гірше, якщо ви використаєте трохи відмінні  початкові умови, ви одержите цілком інші ексцеси (Це і є знаменитий "ефект метелика".  Щоб спостерігати його змінити x(0)=1 у початкових умовах хоча б на  x(0) =1.0000001.). 

Все це виглядаєцілком непередбаченим.  Рівняння, які управляють цією поведінкою, є цілком визначеними і абсолютно простими - як може зовнішня випадковість засмітити щось у цій простій задачі?  Це по справжньому дивовижно, і цей тип поведінки назвали "динамічним хаосом" .  

У пакеті DEtools є й спеціальна більш тонка команда: DEplot. Її синтаксис може бути таким:

DEplot(deqns, vars, trange, xrange, yrange, options)
де deqns- система диференційних рівнянь у числовому форматі, vars- перелік змінних, trange, xrange, yrange - інтервали зміни перемінної та функцій, options - звичайні опції. Наведемо нижче приклад візуалізації вирішення складної фізичної задачі  за допомогою цього інструменту:
> DEtools[DEplot]([diff(x(t),t)=x(t)*(1-y(t)),diff(y(t),t)=.3*y(t)*(x(t)-1)],

[x(t),y(t)],t=-2..2,x=-1..2,y=-1..2,arrows=LARGE,

title=`Модель Лотка-Вольтерра`,color=[.3*y(t)*(x(t)-1),x(t)*(1-y(t)),.12]);

Рішення системи рывнянь презентоване у вигляді векторного поля, стрілки якого є дотичними до кривих рішення (самі ці криві не будуються). Зверніть увагу на функціональне зафарбування стрілок векторного поля, що робить рішення особливо наочнимт на екрані дісплея. 

8.3.  Поглиблений аналіз диференційних рівнянь
Одиночне диференційне рівняння або система диференційних рівнянь називаються автономними, якщо їх права частина явно не залежить від незалежної змінної. Для автономних диференційних рівнянь або систем при побудові графіків рішень функцією DEplot не обов'язково задавати початкові умови, але потрібно указувати діапазон зміни шуканих змінних. Для перевірки рівнянь (або систем) на автономність використовується команда: 

autonomous(des,vars,ivar) 

де des — задане диференційне рівняння або (у вигляді списку) система диференційних рівнянь, vars — залежні змінні; ivar — незалежна змінна. Якщо система автономна, то ця функція повертаєзначення  true, інакше false. 

Наприклад

> with(DEtools):

DE2 := diff(x(u),u)-x(u)*cos(arctan(x(u)))=arctan(u);

> autonomous(DE2,x,u);
[image: image65.wmf]false


Отже наше рівняння не є автономним.

Для поглибленого аналізу аналітичного рішення диференційного рівняння (або системи таких рівнянь) можна використовувати спеціальну можливість підказок щодо типу рівняння. Підказки надає функція odeadvisor 

> DEq:=diff(x(t),t$2) + diff(x(t),t)/tau + omega^2*x(t)=0;

odeadvisor(DEq);

Що треба розуміти так, що перед нами рівняння другого порядку.  Рівняння другого порядку можуть бути типів:

Bessel,      Duffing,       ellipsoidal,      elliptic,    Emden,      

erf,         exact_linear,  exact_nonlinear,  Gegenbauer,  Halm,       

Hermite,     Jacobi,        Lagerstrom,       Laguerre,    Lienard,    

Liouville,   linear_ODEs,   linear_sym,       missing,     Painleve,   

quadrature,  reducible,     sym_Fx,           Titchmarsh,  Van_der_Pol  

Якщо ми вийдемо зараз на гіперпосилку типу missing, то зможемо багато дізнатися щодо того як вирішуються  рівняння саме такого типу.

У багатьох випадках аналітичні рішення навіть простих диференційних рівнянь виявляються вельми складними, наприклад містять спеціальні математичні функції. При цьому нерідко корисною є заміна такого рішення іншим, теж аналітичним, але наближеним рішенням. Найпоширенішим наближеним рішенням в цьому випадку може бути полиномиальное рішення, тобто заміна реального рішення поліномом того або іншого ступеню. При цьому порядок полінома задається значенням системної змінної Order (за умовчанням Order =6), а для отримання такого рішення функція dsolve повинна мати параметр series. Наприклад:

> res1:=dsolve({DE2,x(0)=1},x(u),numeric);

dsolve({DE2,x(0)=1},x(u),series);

Зробимо пару маніпуляцій, щоби порявняти графічно рішення, отримані числовим методом, та наше наближоне рішення у вигляді поліному пятого ступеню:

> res2:=1+1/2*2^(1/2)*u+5/8*u^2+1/48*2^(1/2)*u^3-49/384*u^4+(-23/3840*2^(1/2))*u^5;
[image: image70.wmf] := 

> g1:=plots[odeplot](res1,[u,x(u)],-2..2, color=red):

g2:=plot(res2,u=-2..2, color=navy):
> plots[display](g1,g2, style=point, title=`Точне (червоні точки) та наближене\n (сині точки) рішення DE2`);

Як бачимо в інтервалі  [-1,1] наше наближене рішення виглядає цілком непогано. Перевагою наближеного рішення є те, що з ним можно працювати як з аналітичною функцією, тоді як більш точне числове рішення не є таким наочним.

9. Пакет лінійної алгебри та його можливості
9.1. Головні визначення лінійної алгебри

9.2. Пакет лінійної алгебри . Призначення та завантаження

9.3. Робота з векторами та матрицями

9.4. Рішення систем алгебраїчних рівнянь

9.1. Головні визначення лінійної алгебри

Перш ніж перейти до розгляду великих можливостей пакетів Maple в області рішення задач  лінійної алгебри, розглянемо короткі визначення, що відносяться до неї.  

   Матриця (m х n) - прямокутна двовимірна таблиця, що містить m рядків і n стовпців елементів,  кожний з яких може бути представлений числом, константою, перемінною, символьним або  математичним виразом.

   Квадратна матриця - матриця, у якої число рядків m дорівнює числу стовпців n. 

   Сингулярна (вироджена) матриця - квадратна матриця, у якої детермінант (визначник)  дорівнює 0. Лінійні рівняння з  майже сингулярними матрицями можуть давати великі погрішності при рішенні.  

   Одинична матриця - це квадратна матриця, у якої діагональні елементи рівні 1, а  інші елементи рівні 0. 

   Транспонована матриця - матриця, у  якої стовпці і рядки міняються місцями, тобто елементи транспонованої матриці  задовольняють умові [image: image72.wmf] = 

A

T

(

)

,

i

j

(

)

A

,

i

j

 

   Зворотна матриця - це матриця [image: image73.wmf]M

, що, будучи помножена на вихідну квадратну матрицю М,  дає одиничну матрицю Е.  

   Ранг матриці - найбільший з порядків відмінних від нуля мінорів квадратної матриці.  

   Слід матриці - сума діагональних елементів матриці.  

   Характеристичний багаточлен матриці - визначник різниці цієї матриці й одиничної матриці,  помножений на перемінну багаточлена, - |А - gE|. 
   Власні значення матриці - корені її  характеристичного багаточлена. 

   Матрична форма запису системи лінійних рівнянь - вираз АХ = У, де А - матриця  коефіцієнтів системи, X - вектор невідомих і В - вектор вільних членів. Один зі способів  рішення такої системи очевидна - X = [image: image74.wmf]A

9.2. Пакет лінійної алгебри. Призначення та завантаження

У ядро Maple введені дуже скромні і мінімально необхідні засоби для рішення задач лінійної алгебри. Основний упор у їхній реалізації зроблений на пакети, що підключаються. Основним з них, успадкованим від попередніх реалізацій системи, є пакет рішення задач лінійної алгебри linalg. Це один із самих великих і могутніх пакетів в області рішення задач лінійної алгебри. Він містить понад сто функцій.
   Однак в останні роки розроблювачі систем символьної математики усвідомили, що мала швидкість виконання векторних і матричних операцій при рішенні задач лінійної алгебри обертається втратою помітної частини ринку систем комп'ютерної математики. Так як більшість обчислень у лінійній алгебрі вимагає саме чисельне обчислення, то нові версії таких систем відрізняються від колишніх насамперед різким підвищенням ефективності рішення задач лінійної алгебри в чисельному вигляді. У нових реалізаціях систем Maple  була зроблена ставка на використання давно апробованих швидких алгоритмах лінійної алгебри, запропонованих розробітниками Number Algorithm Group (NAG). Ці алгоритми здавна застосовуються на великих ЕОМ і суперкомп'ютерах, забезпечуючи прискорення чисельних матричних операцій від декількох разів до декількох десятків разів. Їхнє застосування забезпечує ефективне використання систем символьної математики в рішенні задач, що зводяться до задач лінійної алгебри. У числі таких задач численні задачі теоретичної електротехніки, механіки багатьох об'єктів, моделювання електронних пристроїв і т.д. У Maple 8 використання алгоритмів NAG реалізується новим пакетом LinearAlgebra.

Пакет linalg підключається так

> with(linalg);
Warning, the protected names norm and trace have been redefined and unprotected

Пакет LinearAlgebra підключається так

> with(LinearAlgebra);
Warning, the assigned name GramSchmidt now has a global binding

Неважко помітити, що багато функцій цього пакета повторює по призначенню функції більш старого пакета linalg. Ці функції задіють можливості швидких алгоритмів NAG і на відміну від функцій пакета linalg орієнтовані на чисельні розрахунки у форматі обробки речовинних чисел, характерному для комп'ютерної платформи. 

9.3. Робота з векторами та матрицями

У Maple є наступні функції для завдання векторів і матриць:  

vector(n,list) - створення вектора з n елементами, заданими в списку list; 

matrix(n,m,list) - створення матриці з числом рядків n і стовпців m з елементами, заданими списком list.
Або розширені
Vector[o](d, init, ro, sh, st, dt, f, a, o)
де основні з них:
[o] - опція, що задає горизонтальний [row] або вертикальний [column] вектор;
о - теж саме у вигляді orientation=name, 
d - опція, розмір вектора (починаючи з 1)
init - опція, процедура Maple, таблиця, масив, вектор, список та ін. - початкові значення вектора
Matrix(r, c, init, ro, sc, sh, st, o, dt, f, a)
r,c - опція, кількість рядків та стовпців матриці
init - опція, процедура Maple, таблиця, масив, вектор,матриця, список та ін. - початкові значення матриці
> restart:

vector(4,[1,x,x^2,x^3]);

Vector[row]([1, x, x^2, x^3]);

Vector[column]([1, x, x^2, x^3]);

> 

> matrix(2,3,[x,y,z,a,b,c]);

Matrix([[1,2,3],[4,5,6]]);

Matrix(2,3,5);

Matrix(2,3);

Для інтерактивного введення матриць можна, визначивши розмірність деякого масиву, використовувати функцію entermatrix:

> A:=array(1..3,1..2):
Після виконання цього фрагмента документа діалог з користувачем має такий вигляд:

> with(linalg):

entermatrix(A);
Warning, the protected names norm and trace have been redefined and unprotected

enter element 1,1 > 1:
enter element 1,2 > 2:
enter element 2,1 > 3:
enter element 2,2 > 4:
enter element 3,1 > 5:
enter element 3,2 > 6:

> with(LinearAlgebra);

M:=Matrix(6,6,x):

Vr:=Vector[row](6):

Vc:=Vector[column](6):
Для роботи з векторами і матрицями Maple 8 має безліч функцій, що входять у пакет LinearAlgebra. Обмежимося приведенням короткого опису найбільш розповсюджених функцій цієї категорії. 

Операції зі структурою окремого вектора V і матриці М:  

ColumnDimension(M) - повертає число стовпців матриці М;  

> ColumnDimension(M);
[image: image83.wmf]6


RowDimension(M) - повертає число рядків матриці М;

> RowDimension(M);
[image: image84.wmf]6


Dimension(V) - повертає розмірність вектора V або матриці М; 

> Dimension(Vc);

Dimension(Vr);

Dimension(M);
[image: image85.wmf]6


[image: image86.wmf]6

Column(M,і) - повертає i-й стовпець матриці М; 

> Column(M,2);

Row(M,і) - повертає i-ю рядок матриці М; 

> Row(M,3);
[image: image89.wmf][

 Minor(M,і, j) - повертає мінор матриці М для елемента з індексами й і j; 

> Minor(M,2, 2);

DeleteColumn(M,і.. j) - видаляє стовпці матриці М від i-roдо j-ro; 

DeleteRow(M,і..j) - видаляє рядка матриці М від i-й до j-й; 

extend (М, m, n,х) - (пакет linalg)розширює матрицю М на m рядків і n стовпців із застосуванням заповнювача х. 

> DeleteColumn(M,2.. 4);

DeleteRow(M,2..2);

with(linalg):

extend(M, 2, 1, 0); 

Warning, the previous binding of the name GramSchmidt has been removed and it now has an assigned value.
Warning, the protected names norm and trace have been redefined and unprotected

Основні векторні і матричні операції: 

> U:=Vector(3,[1,2,3]):

V:=Vector(3,[6,7,8]):

M:=Matrix(3,3,[[1,2,3],[4,5,6],[7,8,9]]):

M1:=Matrix(3,3,[[11,12,13],[14,15,16],[17,18,19]]):

DotProduct(U,V) - повертає скалярний добуток векторів U і V; 

CrossProduct(U,V) - повертає векторний добуток векторів U і V;

> evalm(DotProduct(U,V));

evalm(CrossProduct(U,V));
[image: image94.wmf]44

]

,

,

-5

10

-5


Norm(V) або Norm(M) - повертає норму вектора або матриці;

> Norm(V);

Norm(M);

[image: image96.wmf]8


[image: image97.wmf]24


copyinto(A,B,і, j) - (пакет linalg) копіює матрицю А в В для елементів послідовно від і до j;

> m:=extend(M1,3,3,0);

copyinto(M,m,4, 4);

[image: image100.wmf]CompanionMatrix


concat(Ml,M2) - (пакет linalg) повертає об'єднану матрицю з горизонтальним злиттям матриць Ml і М2; 

stackmatrix(Ml,M2) - (пакет linalg) повертає об'єднану матрицю з вертикальним злиттям Ml і М2;

> concat(M1,M);

stackmatrix(M1,M);

Add(A,B) і evalm(A+B) - повертає суму матриць А и В; 

Multiply(A,B) і evalm(A&*B) - повертає добуток матриць А и В; 

> Add(M,M1);

Multiply(M,M1);

CharacteristicPolynomial(M,lambda) - повертає характеристичний поліном матриці М щодо заданої перемінної lambda; 

> CharacteristicPolynomial(M,lambda);
[image: image105.wmf] - 

Determinant(M) - повертає детермінант (визначник) матриці М;

> Determinant(M);
[image: image106.wmf]0


Eigenvalues(M,opt) - повертає власні значення матриці М; 

> Eigenvalues(M);

Trace(M) - повертає слід матриці М;

> Trace(M);
[image: image108.wmf]15


Rank(M) - повертає ранг матриці М; 

> Rank(M);
[image: image109.wmf]2


Transpose(M) - повертає транспоновану матрицю М; 

> Transpose(M);
> 

MatrixInverse(M) або evalm(l/M) - повертає матрицю, зворотну до М; 

> M2:=Matrix(2,2,[[1,2],[3,5]]);

MatrixInverse(M2);

9.4. Рішення систем алгебраїчних рівнянь

Кінцевою метою більшості матричних операцій є рішення систем лінійних рівнянь. Для цього пакет LinearAlgebra пропонує безліч методів і засобів їхньої реалізації, один з яких - функція 
LinearSolve(A, B, m, t, c, ip, outopts, methopts) - знаходить рішення рівняння A . x = В
    m - визначення методу (опція) у вигляді method = name де name може бути 'none', 'solve',  'subs', 'Cholesky', 'LU', 'QR', 'SparseLU', 'SparseDirect' або 'SparseIterative'; 
   t  - задає ім'я невизначених змінних у вигляді free = name;
   ip - при зазначенні inplace замінює вектор В розв'язками
> restart:

with(LinearAlgebra):

A := <<1,0,0>|<2,1,0>|<1,0,0>|<-1,-1,-3>>;

b := <2,-1,-9>;

LinearSolve(A, b, free='s');

> P := <<4,2,-1>|<2,4,3>|<-1,3,5>>:

z := <1,0,0>:

LinearSolve(P, z, inplace);

[image: image116.wmf]é

ë

ê

ê

ê

ê

ê

ê

ê

ê

ê

ê

ê

ê

ê

ù

û

ú

ú

ú

ú

ú

ú

ú

ú

ú

ú

ú

ú

ú

11

8

-13

8

5

4


> z;

10. Пакети для наближеного обчислення інтегралів (student)
та інтерполяції кривих (CurveFitting)
10.1. Функції пакету student

10.2. Функції інтегрування в пакеті student

10.3. Функції пакету CurveFitting

10.4. Метод найменших квадратів

10.5. Апроксимація сплайнами, безперервними дробами, поліномами

10.1. Функції пакету student

Пакет student - це, безсумнівно, один з пакетів, найбільш привабливих для студентів і аспірантів. У ньому зібрані найбільш розповсюджені і потрібні функції, що студенти університетів і інших вузів звичайно використовують на практичних заняттях, при підготовці курсових і дипломних проектів. У Maple 8 пакет має наступні функції (показані після виклику пакету):

> with(student);
[image: image118.wmf]D

Це найбільш розповсюджені функції, в основному стосовні до математичного аналізу.

У цьому пакеті в попередній версії системи (Maple 7) було майже півсотні функцій. Нижче представлена більшість з них:

D - диференційний оператор;

Diff - інертна форма функції обчислення похідній;

Doubleint - інертна форма функції обчислення подвійного інтеграла;

Int - інертна форма функції інтегрування int;

Limit - інертна форма функції обчислення межі limit;

Lineint - інертна форма функції обчислення лінійного інтеграла lineint;

Point - тестування об'єкта на відповідність типу точки (point);

Product - інертна форма функції обчислення добутку членів послідовності;

Sum - інертна форма функції обчислення суми членів послідовності;

Tripleint - інертна форма функції обчислення потрійного інтеграла;

changevar - заміна змінної;

combine - об'єднання подібних членів;

completesquare - обчислення повного квадрата (багаточлена);

distance - обчислення відстані між точками;

equate - створення системи рівнянь зі списків, таблиці, масивів;

extrema - обчислення екстремуму виразу;

integrand - повертає підінтегральний вираз з-під знака інертного інтегралу;

intercept - знаходження крапки перетинання двох кривих;

intparts - інтегрування по частинах;

isolate - виділення підвиразу;

leftbox - графічна ілюстрація інтегрування методом лівих прямокутників;

leftsum - числове наближення до інтеграла лівими прямокутниками;

makeproc - перетворення виразу в процедуру Maple;

maximize - обчислення максимуму функції;

middlebox - графічна ілюстрація інтегрування методом центральних прямокутників;

middlesum - числове наближення до інтеграла центральними прямокутниками;

midpoint - обчислення середньої точки сегмента лінії;

minimize - обчислення мінімуму функції;

powsubs - підстановка для множників виразу;

rightbox - графічна ілюстрація інтегрування методом правих прямокутників;

rightsum - числове наближення до інтеграла правими прямокутниками;

showtangent - графік функції і дотичної лінії;

simpson - числове наближення до інтеграла по методу Симпсона;

slope - обчислення і побудова дотичної до заданої точки функції;

trapezoid - числове наближення до інтеграла методом трапецій;

value - обчислює інертні функції.

> distance([a, b], [c, d]);

> a:=[x+y,x-y];

b:=[2,3];

equate(a,b);

> showtangent(x^2+5, x = 2);

> p := (x+2)^3 + 1/(x+2) + sin(x+2);

powsubs(x+2 = z, p);

10.2. Функції інтегрування в пакеті student

 У ядрі й у пакетах розширення Maple 8 можна знайти безліч спеціальних функцій для обчислення інтегралів різного типу. Наприклад, у пакеті student маються наступні функції:

Int(expr, x) - інертна форма обчислення невизначеного інтеграла;

Doubleint(expr,x,y,Domain) - обчислення подвійного інтеграла по змінним х и у по області Domain;

Tripleint(expr,x,y,z) - обчислення потрійного інтеграла;

intparts(f,u) - інтегрування за частинами.
> with(student):

intparts(Int(x^k*ln(x), x), ln(x));

Також у пакеті student є засоби чисельного інтегрування. Наприклад, метод лівих прямокутників:
leftbox(f(x), x=a..b, n, 'shading'=<color>, <plot options>) де n - кількість прямокутників

leftsum(f(x), x=a..b,n)
> leftbox(sin(x)*x+sin(x), x=0..2*Pi, 5, shading=BLUE);

leftsum(sin(x)*x+sin(x), x=0..2*Pi,5);

evalf(%);

[image: image127.png]

[image: image129.wmf]-5.433738024


Аналогічно використовуються інші наближення інтегрування прямокутниками.
Також можна використати інтегрування методом трапецій
trapezoid(f(x), x=a..b, n)
> trapezoid(x^3*ln(x), x=1..3);

evalf(%);

[image: image131.wmf]18.03096539


або методом Сімпсона

simpson(f(x), x=a..b, n)
> simpson(x^3*ln(x), x=1..3);

evalf(%);

[image: image133.wmf]17.24913939


10.3. Функції пакету CurveFitting

Новий пакет наближення кривих CurveFitting досить корисний тим, хто займається досить розповсюдженою задачею, як наближення кривих. Він містить ряд функцій: 

> with(CurveFitting);

Доступ до функцій пакета можливий за допомогою конструкцій: 

CurveFitting[function](arguments)  

function(arguments)
Функції пакету: 

BSpline - обчислює В-сплайни

BsplineCurve - побудова B-cплайнів

LeastSquares - реалізація апроксимації за методом найменших квадратів

PolynomialInterpolation - реалізація поліноміальної інтерполяції

RationalInterpolation - раціональна інтерполяція

Spline - обчислення звичайних сплайнів

ThieleInterpolation - апроксимація безперервними дробами

10.4. Метод найменших квадратів

Функція LeastSquares служить для реалізації апроксимації по методу найменших квадратів: 

LeastSquares (xydata, v, opts)  

LeastSquares(xdata, ydata, v.,opts)
де
xydata - список, масив або матриця точок у формі [[xl.yl], [х2,у2],..., [хn,уn]]; 

xdata - список, масив або вектор значень незалежної змінної [xl,x2,... ,хn]; 

ydata - список, масив або вектор значень залежної змінної у формі [у1,у2,...,уn]; 

v - ім'я незалежної змінної; 

opt - необов'язкові опції weight=wlist, curve=f або params=pset
> LeastSquares([[0,1],[1,2],[2,3],[3,10]], v);

> restart:

with(CurveFitting):

LeastSquares([0,1,3,5,6], [2,-1,-3,6,8], x, curve=a*x^2+b*x+c);
[image: image136.wmf] - 

10.5. Апроксимація сплайнами, безперервними дробами, поліномами
         Функція обчислення звичайних сплайнов Spline 

Функція: 

Spline(xydata, v, dgr, endpts) 

Spline(xdata, ydata, v, opts, dgr, endpts) 
обчислює звичайні (не В-типу) сплайни
dgr - степінь сплайну, degree=d
endpts - тип граничних умов

> Spline([[0,0],[1,1],[2,4],[3,3]], v);
[image: image137.wmf]ì

> Spline([0,1,2,3], [0,1,4,3], v, degree=1);
[image: image138.wmf]ì

Функція апроксимації безперервними дробами ThieleInterpolation 
Функція ThieleInterpolation здійснює інтерполяцію на основі безперервних дробів. Вона задається у виді: 

ThieleInterpolation (xydata, v) 
ThieleInterpolation(xdata, ydata, v) 
> ThieleInterpolation([[1,3],[2,4],[4,5],[5,8]], v);

Функція поліноміальної апроксимації PolynomialInterpolation 
Функція PolynomialInterpolation реалізує поліноміальну інтерполяцію і може використовуватися у виді: 

PolynomialInterpolation (xydata, v)  

PolynomialInterpolation(xdata, ydata, v) 

v - може бути як змінною, так і будь-яким виразом
> PolynomialInterpolation([[0,0],[1,3],[2,1],[3,3]], z);
[image: image140.wmf] - 

> PolynomialInterpolation([[0,0],[1,3],[2,1],[3,3]], 1);
[image: image141.wmf]3

